

บทที่ 6

โปรแกรมคอมพิวเตอร์

เนื่องจากโปรแกรมเขียนบนไมโครคอมพิวเตอร์ที่ขนาดค่อนข้างเล็ก (Apple II) และมีการคำนวณซับซ้อนทั้งใช้ตัวแปรเป็นจำนวนมาก จึงได้เขียนโปรแกรมแยกเป็น 2 โปรแกรม

โปรแกรมแรกเป็นการรับข้อมูลค่าคงที่ของเครื่องมือและตำแหน่งจุดอ้างอิงและโปรแกรมสร้างวิธีเปลี่ยนระบบพิกัดจากการวัด (U, V) เป็นพิกัดของห้องฟอง (x, y)

โปรแกรมที่ 2 ใช้ข้อมูล บางส่วนจากโปรแกรมแรก และรับข้อมูลใหม่คือตำแหน่งบนภาพทางเดินอนุภาคจากภาพถ่ายแล้ววิเคราะห์หาข้อมูลสำคัญของทางเดินอนุภาค

6.1 โปรแกรมสร้างวิธีเปลี่ยนระบบพิกัด

คือลำดับขั้นการทำงาน A และ B ในรูปที่ 12 ทำงานตามขั้นตอน 5.1 ซึ่งแยกการทำงานเป็นขั้นตอนย่อยต่าง ๆ ดังนี้

6.1.1 การรับข้อมูล และตรวจสอบการป้อนข้อมูลค่าคงที่และจุดอ้างอิง มีขั้นตอนการทำงานดังนี้

1. จองตัวแปรสำหรับเก็บข้อมูล
2. พิมพ์ข้อความบอกผู้ใช่ว่าจะรับข้อมูลใด
3. รับข้อมูลตาม 2 ซึ่ง เป็นลำดับตามแบบที่ให้ไว้ ท้ายบทที่ 4 ข้อ 1-17
4. ตรวจสอบและแก้ไขข้อมูลที่ผิดพลาด

6.1.1.1 ความหมายของตัวแปร

6.1.1.1.1 จุดอ้างอิง

ตัวแปร (FX (CN, I), FY (CN, I))

F หมายถึง Fiducial Mark

x, y, z หมายถึง ระบบของห้องฟอง

ตัวแปร CN หมายถึง เลขหมายของภาพหรือกล้อง

ตัวแปร I หมายถึง ลำดับที่ของจุดอ้างอิง

ตัวแปร FU (CN, I), FV (CN, I)

U, V หมายถึง ระบบของเครื่องวัดตำแหน่งจากภาพ

6.1.1.1.2 ค่าคงที่ของเครื่องมือ

ตัวแปร W หมายถึง ระยะ B_1B_2 (ตาม 4.3.1.2.1)

ตัวแปร U_0 หมายถึง ตำแหน่ง B_1 ค่า U_0 (ตาม 4.3.1.2.2)

ตัวแปร V_0 หมายถึง ตำแหน่ง B_2 ค่า V_0 (ตาม 4.3.1.2.3)


ตัวแปร FZ(1) หมายถึง ระนาบ B (ตาม 4.3.1.1.1)

ตัวแปร FZ(2) หมายถึง ระนาบ A (ตาม 4.3.1.1.2)

ตัวแปร CH# เป็นตัวแปรรับการตรวจสอบการป้อนข้อมูล


6.1.1.2 ผังการทำงานและโปรแกรม

ผังงาน



รูปที่ 30

ลำดับการทำงานตาม 6.1.1



โปรแกรม

```

10 HOME
20 PRINT : PRINT "THIS PROGRAM C
  AN FIT HELIX TRACKS."
30 :PI = 3.141592653
80 GOTO 10070: REM MAIN PROGRAM
  
```

```

10000 REM *****
10010 REM *
10020 REM * MAIN PROGRAM *
10030 REM *
10040 REM *****
10050 REM SET VARIABLES FOR
10060 REM FIDUCIAL POINTS
10070 DIM FX(2,3),FY(2,3),FZ(2)
10080 DIM FU(2,3),FV(2,3)
10090 REM *****
  
```

รูปที่ 31

โปรแกรมคอมพิวเตอร์ที่ทำงานตาม 6.1.1

```

11420 PRINT
11430 PRINT "DISTANCE BETWEEN B1
AND B2"
11440 INPUT "IS W = ";W
11450 INPUT "POSITION B1 U = ";U
O
11460 INPUT "POSITION B2 V = ";V
O
11490 REM *****
11500 HOME : REM CHECKING
11510 PRINT TAB( 18)"NOW,"
11520 PRINT
11530 PRINT "I WILL SHOW YOU EAC
H VALUE WAS INPUT."
11540 PRINT
11550 : PRINT "IF IT IS THE ";:
INVERSE : PRINT "RIGHT";: NORMAL : PRINT :
" VALUE, YOU PRESS ";: INVERSE
: PRINT "R";: NORMAL : PRINT
"."
11560 : NORMAL : PRINT : PRINT "I
F";: PRINT TAB( 14);: INVERSE
: PRINT "WRONG";: NORMAL : PRINT
TAB( 31)"PRESS ";: INVERSE
: PRINT "W";: NORMAL : PRINT
"."
11570 PRINT : PRINT "PREPARE DAT
A FOR CHECKING."
11580 PRINT : PRINT TAB( 10)"PR
ESS ANY KEY WHEN READY.";: GET
CH$
11590 HOME
11600 PRINT
11610 INVERSE : HTAB 7: PRINT "R
";: HTAB 10: PRINT "=";: HTAB
13: PRINT "RIGHT";: HTAB 26:
PRINT "W";: HTAB 29: PRINT
 "=";: HTAB 32: PRINT "WRONG"
: NORMAL
11620 PRINT
11630 PRINT "FIDUCIAL POINTS DAT
A"
11640 PRINT : PRINT "FIDUCIAL PL
ANES"
11650 PRINT "Z1 = "FZ(1);: GET C
H$: IF CH$ = "W" THEN 61650
11660 HTAB (20): PRINT "Z2 = "FZ
(2);: GET CH$: IF CH$ = "W" THEN
61660

```

```

10120 PRINT TAB( 15)"FIRST TIME
"
10130 PRINT " THIS PROGRAM WIL
L HELP YOU TRANSFORM"
10140 PRINT "MEASURED COORDINATE
(U,V) TO THE"
10150 PRINT "COORDINATE OF FIDUC
IAL MARKS (X,Y)."

```

```

11670 FOR CN = 1 TO 2
11680 PRINT : PRINT : PRINT "VIE
      W "CN
11690 : FOR I = 1 TO 3
11700 : PRINT : PRINT "POINT "I
11710 : PRINT " X = ";FX(CN,I);: GET
      CH$: IF CH$ = "W" THEN 61710

11720 : HTAB (20): PRINT " Y = ";
      FY(CN,I);: GET CH$: IF CH$ =
      "W" THEN 61720
11730 : PRINT : PRINT " U = ";FU(
      CN,I);: GET CH$: IF CH$ = "W
      " THEN 61730
11740 : HTAB (20): PRINT " V = ";
      FV(CN,I);: GET CH$: IF CH$ =
      "W" THEN 61740
11750 : NEXT I
11755 PRINT
11760 NEXT CN
11770 PRINT : PRINT "DISTANCE BE
      TWEEN B1 AND B2 IS W = "W;: GET
      CH$: IF CH$ = "W" THEN 61770

11780 PRINT : PRINT "POSITION OF
      B1 U = "U;: GET CH$: IF C
      H$ = "W" THEN 61780
11790 PRINT : PRINT "POSITION OF
      B2 V = ";V;: GET CH$: IF C
      H$ = "W" THEN 61790
11810 PRINT : PRINT "ALL RIGHT ?
      (Y FOR YES, N FOR NO)";
11820 GET CH$
11830 IF CH$ = "Y" THEN 11910
11840 IF CH$ = "N" THEN 11590
11850 PRINT "";: GOTO 11820
11860 REM END OF CHECKING F.P.

```

```
61630 REM, *****
61640 REM CORRECTING INPUT F.P.
61650 PRINT : PRINT "CORRECT Z1
      FROM "FZ(1)" TO BE";: INPUT
      FZ(1): GOTO 11590
61660 PRINT : PRINT "CORRECT Z2
      FROM "FZ(2)" TO BE";: INPUT
      FZ(2): GOTO 11590
61710 PRINT : PRINT "CORRECT X =
      "FX(CN,I);: INPUT " TO BE ";
      FX(CN,I): GOTO 11590
61720 PRINT : PRINT "CORRECT Y =
      "FY(CN,I);: INPUT " TO BE "
      ;FY(CN,I): GOTO 11590
61730 PRINT : PRINT "CORRECT U =
      "FU(CN,I);: INPUT " TO BE "
      ;FU(CN,I): GOTO 11590
61740 PRINT : PRINT "CORRECT V =
      "FV(CN,I);: INPUT " TO BE "
      ;FV(CN,I): GOTO 11590
61770 PRINT : PRINT "CORRECT W =
      "W;: INPUT " TO BE ";W: GOTO
      11590
61780 PRINT : PRINT "CORRECT UO
      = ";UO;: INPUT " TO BE ";UO:
      GOTO 11590
61790 PRINT : PRINT "CORRECT VO
      = ";VO;: INPUT " TO BE ";VO:
      GOTO 11590
61820 REM READY CORRECTING F.P.
61830 REM *****
```

ศูนย์วิทยทรัพยากร
จุฬาลงกรณ์มหาวิทยาลัย

6.1.2 เปลี่ยนระบบ (U,V) เป็น (X,Y) มี B₁ เป็นจุด (0,0) มีขั้นตอนการทำงานดังนี้

1. จองตัวแปรสำหรับรับข้อมูลที่ถูกละเปลี่ยน
2. เปลี่ยนค่า (U,V) ให้เป็นระยะ (B₁B₀, B₂B₀) (ตาม 5.1.1)

และเปลี่ยนสเกล (หารด้วย 40)

3. แปลงค่า (U,V) หลังแปลงตาม 2 แล้วให้เป็น (X,Y) ที่ละคู่ จนกว่าจะครบเก็บในตัวแปรในข้อ 1

6.1.2.1 ความหมายของตัวแปร

6.1.2.1.1 จุดอ้างอิง

ตัวแปร X1(CN,I), Y1(CN,I)

X,Y หมายถึงพิกัดที่มี B₁ เป็นจุดกำเนิด

1 หมายถึงการแปลงครั้งที่ 1

6.1.2.1.2 โปรแกรมย่อย

ตัวแปร SO หมายถึงครึ่งผลบวกด้านทั้ง 3 ของสามเหลี่ยม

ตัวแปร SW = SO - W

ตัวแปร SU = SO - U



ตัวแปร SV = SO - V


ตัวแปร X หมายถึงพิกัด X ตามสมการ (5.4)

ตัวแปร Y หมายถึงพิกัด Y ตามสมการ (5.5)

6.1.2.2 ผังการทำงานและโปรแกรม

ผังงาน





รูปที่ 32

ลำดับการทำงานเปลี่ยนระบบพิกัด (U,V) เป็น (X,Y)

โปรแกรมย่อย

```

90 REM *****
100 REM SUBROUTINE UV TO XY
110 SO = (U + V + W) / 2
120 SW = SO - W
130 SU = SO - U
140 SV = SO - V
150 X = 2 * SO * SV / W - U
160 Y = - 2 * SQR (SO * SU * SV
    * SW) / W
170 RETURN
180 REM END OF UV TO XY
181 REM *****
  
```

รูปที่ 33

โปรแกรมการเปลี่ยนพิกัด (U,V) เป็น (X,Y)

โปรแกรมหลัก

```

10170 DIM X1(2,3),Y1(2,3)
10180 DIM X2(2,3),Y2(2,3)
10190 DIM X0(2),Y0(2),Z(2)
10191 REM TRANSLATION VARIABLES
10200 DIM XL(2),YL(2)
10201 REM LENS POSITIONS
10210 DIM RA(2)
10211 REM ROTATION ANGLE
11020 REM FIRST PART TRANSFORM
11030 REM COORDINATE (U,V) FROM
11040 REM PICTURE MEASUREMENT
11050 REM TO COORDINATE OF THE
11060 REM FIDUCIAL POINTS (X,Y)
11920 FOR CN = 1 TO 2
11930 FOR I = 1 TO 3
11940 ::FU(CN,I) = FU(CN,I) - U0
11950 ::FV(CN,I) = V0 - FV(CN,I)
11960 ::FU(CN,I) = FU(CN,I) / 40
11970 ::FV(CN,I) = FV(CN,I) / 40
11980 NEXT I
11990 NEXT CN
12000 HOME
12010 HTAB 10: VTAB 10: PRINT "F
LEASE WAIT FOR 1 MINUTES."
12020 W = W / 40
12050 REM CN = CAMERA: I = POINT
12060 FOR CN = 1 TO 2
12070 : FOR I = 1 TO 3
12080 :U = FU(CN,I):V = FV(CN,I)
12090 : GOSUB 110
12100 :X1(CN,I) = X:Y1(CN,I) = Y
12110 : NEXT I
12120 NEXT CN
12180 REM END OF (U,V) TO (X,Y)

```

รูปที่ 34

โปรแกรมเปลี่ยนพิกัด (U,V) เป็น (X,Y)

6.1.3 เปลี่ยนระบบ (X,Y) ที่มี B_1B_2 เป็นแกน X โดยหมุนรอบแกน Z ให้เป็นระบบที่มีแกน X สนานกับแกนของระบบพิกัดของห้องฟอง

มีขั้นตอนการทำงานดังนี้


1. จองตัวแปรสำหรับรับข้อมูลที่ถูกลมุแกน
2. จองตัวแปรสำหรับเก็บมุมที่ใช้หมุนแกน
3. คำนวณมุมที่หมุนแกนตาม 5.1.3
4. เก็บค่ามุมที่หมุนแกนจากภาพแต่ละภาพในตัวแปร

6.1.3.1 ความหมายของตัวแปร

ตัวแปร	RA (CN)	ใช้เก็บมุมในหน่วยเรเดียนที่หมุนตามหมายเลขกล้อง (CN)
ตัวแปร	TH	หมายถึงมุมที่หมุนใช้ในการคำนวณในโปรแกรมย่อยตามสมการ (5.8)
ตัวแปร	DX	หมายถึงความแตกต่างบนพิกัด x
ตัวแปร	DY	หมายถึงความแตกต่างบนพิกัด y
ตัวแปร	PI	หมายถึงค่า π (3.141592653)

6.1.3.2 ผังการทำงานและ โปรแกรม


ผังงานย่อยคำนวณมุม



รูปที่ 35

ลำดับการทำงานคำนวณมุม


ผังงานย่อยหมุนแกน



รูปที่ 36


ลำดับการทำงานหมุนแกน

ผังงานหลัก



รูปที่ 37

โปรแกรมคำนวณมุมและหาพื้นที่



รูปที่ 37 (ต่อ)

โปรแกรมย่อย

```

200 IF DX = 0 THEN TH = PI / 2: GOTO
    230
210 :TH = ATN (DY / DX): IF DX >
    0 THEN 240
220 :TH = TH + PI: GOTO 240
230 IF DY < 0 THEN TH = - PI /
    2
240 RETURN
250 REM END OF ANGLE
  
```

รูปที่ 38

โปรแกรมย่อยคำนวณมุม

```

300 :X2 = X1 * COS (RA(CN)) - Y1
      * SIN (RA(CN))
310 :Y2 = X1 * SIN (RA(CN)) + Y1
      * COS (RA(CN))
320 RETURN
330 REM END OF ROTATION

```

รูปที่ 39

โปรแกรมย่อยหมุนแกน

โปรแกรมหลัก

```

12190 REM *****
12195 REM * ROTATION ANGLE *
12196 REM *****
12200 FOR CN = 1 TO 2
12210 :DX = FX(CN,2) - FX(CN,1)
12220 :DY = FY(CN,2) - FY(CN,1)
12230 GOSUB 200: REM * ALPHA *
12240 RA(CN) = TH
12250 :DX = X1(CN,2) - X1(CN,1)
12260 :DY = Y1(CN,2) - Y1(CN,1)
12270 GOSUB 200: REM ** BETA **
12280 RA(CN) = RA(CN) - TH
12290 NEXT CN
12300 FOR CN = 1 TO 2
12310 : FOR I = 1 TO 3
12330 :::X1 = X1(CN,I)
12340 :::Y1 = Y1(CN,I)
12350 ::: GOSUB 300: REM ROTATE
12360 :::X2(CN,I) = X2
12370 :::Y2(CN,I) = Y2
12380 : NEXT I
12390 NEXT CN
12400 REM END OF ROTATION

```

รูปที่ 40

โปรแกรมหลักหมุนแกน

6.1.4 เลื่อนแกนให้เป็นระบบเดียวกับระบบพิกัดในห้องฟอง
มีขั้นตอนดังนี้


1. จงตัวแปรสำหรับจุดกำเนิดแกนที่เลื่อน
2. จงตัวแปรสำหรับระนาบของภาพที่ทำการวัด
3. จงตัวแปรสำหรับตำแหน่งเลนส์และสำหรับสมการเชิงเส้น
4. คำนวณระนาบของภาพที่ทำการวัดของแต่ละภาพ
5. คำนวณการเลื่อนแกนและตำแหน่งเลนส์ของแต่ละภาพ
6. เก็บข้อมูลตำแหน่งการเลื่อนแกนและตำแหน่งเลนส์ของแต่ละภาพ

6.1.4.1 ความหมายของตัวแปร

ตัวแปร $XO(I), YO(I)$	หมายถึงพิกัด (X, Y) ของการเลื่อนแกนไป ของภาพในกล้องที่ I
ตัวแปร $Z(I)$	หมายถึงระนาบของภาพของกล้องที่ I
ตัวแปร $XL(I), YL(I)$	หมายถึงพิกัด (X, Y) ของตำแหน่งเลนส์ของกล้องที่ I
ตัวแปร $DO, D1, D2$	หมายถึงตัวแปรที่ใช้ช่วยแก้ระบบสมการเชิงเส้น 2 ตัวแปร
ตัวแปร $A(I, J)$	หมายถึงสัมประสิทธิ์ของตัวไม่ทราบค่าในสมการเชิงเส้น

6.1.4.2 ตั้งงานและโปรแกรม


ตั้งงานย่อย



รูปที่ 41

ตั้งงานย่อยคำนวณค่าดีเทอร์มิแนนต์

ผังงานหลัก



รูปที่ 42

ลำดับการทำงานปรับระบบแกนเดียวกับแกนของห้องฟอง

โปรแกรมย่อย

```

340 REM *****
350 REM DETERMINANT 2*2
360 :D0 = A(2,2) - A(1,2)
370 :D1 = A(1,1) * A(2,2) - A(1,2)
    :      * A(2,1)
380 :D2 = A(2,1) - A(1,1)
390 RETURN
400 REM END OF DETERMINANT
410 REM *****
  
```

รูปที่ 43

โปรแกรมย่อยคำนวณค่าดีเทอร์มิแนนต์

โปรแกรมหลัก

```

12130 FOR I = 1 TO 2
12140 Z(I) = ((X1(I,2) - X1(I,1))
           ^ 2 + (Y1(I,2) - Y1(I,1)) ^
           2) / ((FX(I,2) - FX(I,1)) ^
           2 + (FY(I,2) - FY(I,1)) ^ 2)

12150 Z(I) = SQRT(Z(I)) * FZ(I)
12160 NEXT I
12410 REM *****
12411 REM LENS POSITION AND
12412 REM TRANSLATION X AXIS
12413 REM *****
12420 DIM A(2,2)
12430 FOR CN = 1 TO 2
12440 A(1,2) = Z(CN) / FZ(1)
12450 A(1,1) = A(1,2) * FX(CN,1) -
           X2(CN,1)
12460 A(1,2) = A(1,2) - 1
12470 A(2,2) = Z(CN) / FZ(2)
12480 A(2,1) = A(2,2) * FX(CN,3) -
           X2(CN,3)
12490 A(2,2) = A(2,2) - 1
12500 GOSUB 360: REM DETER 2X2
12510 X0(CN) = D1 / D0
12520 XL(CN) = D2 / D0
12529 REM *****
12530 REM #TRANSLATION Y AXIS #
12531 REM *****
12540 A(1,2) = Z(CN) / FZ(1)
12550 A(1,1) = A(1,2) * FY(CN,1) -
           Y2(CN,1)
12560 A(1,2) = A(1,2) - 1
12570 A(2,2) = Z(CN) / FZ(2)
12580 A(2,1) = A(2,2) * FY(CN,3) -
           Y2(CN,3)
12590 A(2,2) = A(2,2) - 1
12600 GOSUB 360: REM DETER 2X2
12610 Y0(CN) = D1 / D0
12620 YL(CN) = D2 / D0
12630 NEXT CN
12640 REM *****
12650 REM END OF TRANSLATION

```

6.1.5 เก็บข้อมูลที่ใช้จัดการทางเดินอนุภาคลงในงานบันทึกแม่เหล็ก มีขั้นตอนดังนี้

1. ตั้งชื่อแฟ้มข้อมูล (ควรตั้งให้มีความหมาย เช่น หมายเลขฟิล์ม)
2. เปิดแฟ้มข้อมูล ชื่อ "NAME"
3. บรรจุข้อมูล ชื่อแฟ้มข้อมูลตามข้อ 1
4. ปิดแฟ้มข้อมูล ชื่อ "NAME"
5. เปิดแฟ้มข้อมูลชื่อตาม 1
6. บรรจุข้อมูลที่จำเป็นตามลำดับ
7. ปิดแฟ้มข้อมูลชื่อตาม 1
8. จบโปรแกรมตามหัวข้อ 6.1


6.1.5.1 ความหมายของตัวแปร

ตัวแปร NA\$ เก็บชื่อที่ตั้งตามข้อ 1

ตัวแปรอื่น ๆ มีความหมายตามเดิมดังที่กล่าวในหัวข้อก่อน ๆ

6.1.5.2 ผังงานและโปรแกรม

ผังงาน



รูปที่ 45

ลำดับการเก็บข้อมูลลงงานบันทึกแม่เหล็ก

โปรแกรม

```

13010 PRINT
13020 PRINT " PLEASE NAME THIS TRACK.";
13022 VTAB 5: HTAB 10: PRINT "PLEASE WAIT !!!"
13030 INPUT " ?";NA$
13040 PRINT "OPEN NAME"
13050 PRINT "DELETE NAME"
13060 PRINT "CLOSE NAME"
13070 PRINT "OPEN NAME"
13080 PRINT "WRITE NAME"
13090 PRINT NA$
13100 PRINT "CLOSE NAME"
13110 PRINT "OPEN";NA$
13120 PRINT "DELETE";NA$
13130 PRINT "OPEN "NA$
13140 PRINT "WRITE "NA$
13150 PRINT W
13153 PRINT UO
13156 PRINT VO
13170 PRINT ""
13180 FOR I = 1 TO 2
13190 PRINT "WRITE "NA$
13200 PRINT RA(I)
13210 PRINT FZ(I)
13220 : PRINT Z(I)
13230 PRINT XL(I)
13240 PRINT YL(I)
13250 PRINT XO(I)
13260 PRINT YO(I)
13270 PRINT ""
13280 NEXT I
13290 PRINT "CLOSE "NA$
13300 PRINT "RUN FITT"
60000 END

```

รูปที่ 46

โปรแกรมบันทึกข้อมูลลงจากแม่เหล็ก

6.2 โปรแกรมวิเคราะห์ทางเดินอากาศ

คือลำดับขั้นการทำงาน C-I ของผังงานในรูปที่ 12 มีทฤษฎีและวิธีคำนวณตามหัวข้อ 5.2 และ 5.3

6.2.1 การรับข้อมูลในส่วนที่คำนวณในหัวข้อ 5.1 จากแผ่นบันทึก มีขั้นตอนการทำงานดังนี้

1. จองตัวแปรสำหรับรับข้อมูล
2. เปิดเพิ่มข้อมูลชื่อ "NAME"
3. อ่านเพิ่มข้อมูล "NAME" เก็บในตัวแปร NA\$
4. ปิดเพิ่มข้อมูลชื่อ "NAME"
5. เปิดเพิ่มข้อมูลชื่อตามตัวแปร NA\$
6. อ่านข้อมูลตามลำดับเก็บในตัวแปรที่เหมาะสม
7. ปิดเพิ่มข้อมูลชื่อตามตัวแปร NA\$

6.2.1.1 ความหมายของตัวแปร

มีเพียงตัวแปรเดียวที่ต่างไปจากเดิมคือ


ตัวแปร B1 หมายถึง ตัวแปร Uo นั้นเอง และ

ตัวแปร B2 หมายถึง ตัวแปร Vo

ศูนย์วิทยพัทธยากร
จุฬาลงกรณ์มหาวิทยาลัย

6.2.1.2 ผังงานและโปรแกรม

ผังงาน



รูปที่ 47

ลำดับการเรียกข้อมูลจากแผ่นบันทึกแม่เหล็ก

โปรแกรม

```

19999 REM *****
20000 REM READ TEXT FILE
20001 REM *****
20010 DIM Z(2),FZ(2),RA(2)
20020 DIM XO(2),YO(2)
20030 DIM XL(2),YL(2)
20050 PRINT "OPEN NAME"
20060 PRINT "READ NAME"
20070 INPUT NA$
20080 PRINT "CLOSE NAME"
20090 PRINT "OPEN "NA$
20100 PRINT "READ "NA$
20110 INPUT W
20113 INPUT B1
20116 INPUT B2
20130 PRINT ""
20140 FOR I = 1 TO 2
20150 PRINT "READ "NA$
20160 INPUT RA(I)
20170 INPUT FZ(I)
20180 : INPUT Z(I)
20190 INPUT XL(I)
20200 INPUT YL(I)
20210 INPUT XO(I)
20220 INPUT YO(I)
20230 PRINT ""
20240 NEXT I
20250 PRINT "CLOSE "NA$

```

รูปที่ 48

โปรแกรมอ่านข้อมูลจากงานบันทึกแม่เหล็ก

6.2.2 การรับข้อมูล และตรวจสอบข้อมูลตำแหน่งบนทางเดินอนุภาค มีขั้นตอนดังนี้

1. จองตัวแปรเพื่อเก็บข้อมูลจากแป้นพิมพ์
2. รับข้อมูลเรียงลำดับตามภาพ และจุดจากทางเดินจนครบ
3. ตรวจสอบข้อมูลว่ามีการป้อนข้อมูลผิดพลาดหรือไม่ ถ้าผิดให้แก้ไขใหม่

6.2.2.1 ความหมายของตัวแปร


ตัวแปร TU(CN,I), TV(CN,I)

T หมายถึงทางเดินอนุภาค (Track)

U,V หมายถึงพิกัดในการวัดจากเครื่องวัด

ตัวแปร VT เป็นตัวแปรที่ใช้กำหนดตำแหน่งในการพิมพ์ข้อความบนจอภาพ (VTAB)

6.2.2.2 ผังงานและโปรแกรม



รูปที่ 49

ลำดับขั้นการรับข้อมูลทางเดินอนุภาคจากแป้นพิมพ์และการตรวจสอบการป้อนข้อมูล

โปรแกรม

```

20255 GOTO 20601
20260 REM END OF TEXT FILE
20270 REM *****
20300 REM INPUT FROM KEYBOARD
20301 REM *****
20310 HOME
20320 PRINT : PRINT : PRINT "PLEASE KEY DATA OF STUDYING TRACK."
20340 PRINT : PRINT "I CAN FIT CURVE FROM VIEW1 BY POLYNOMIAL REGRESSION."
20350 PRINT "AND CREATE CONICAL SURFACE BY LENS POSITION AS THE VERTEX."
  
```

รูปที่ 50

โปรแกรมรับข้อมูลตรวจสอบการป้อนข้อมูลและแก้ไข

```

20360 PRINT "THEN DRAW STRIGHT L
      INES PASS POSITIONS ON TRAC
      K IN VIEW2 AND LENS POSI
      TION CUT THE SURFACE GENE
      RATED BY VIEW1."
20370 PRINT "FINALLY I CAN MAKE
      A HELIX FIT FROM THE INTERSE
      CTIONS."
20380 PRINT : PRINT "BECAREFUL I
      NPUTTING DATA MUST RESPECTIV
      E THE DATA OF FIDUCIAL POINT
      S."
20390 PRINT : PRINT "FROM EACH V
      IEW, YOU MUST READ OUT SEVEN
      POINTS ALONG THE TRACK IN (
      U,V) COORDINATE."
20400 PRINT : PRINT "PRESS ANY K
      EY WHEN READY.";
20410 GET A$: PRINT
20411 REM *****
20420 REM INPUT DATA BY K.B.
20430 REM READING FROM TRACK
20440 DIM TU(2,6),TV(2,6)
20500 FOR CN = 1 TO 2
20510 PRINT : PRINT "DATA FROM V
      IEW NUMBER ";; FLASH : PRINT
      CN: NORMAL
20520 PRINT
20530 : FOR I = 0 TO 6
20540 : PRINT "POSITION NUMBER ";;
      I + 1
20550 : INPUT " U = ";TU(CN,I)
20560 : INPUT " V = ";TV(CN,I)
20570 : NEXT I
20580 NEXT CN
20590 REM END TRACK READING
20591 REM *****
20600 GOSUB 6000: REM CHECKING

```




```

60000 HOME : PRINT : PRINT : PRINT
      " THIS IS THE CHECKING PA
      RT."
60010 PRINT : PRINT " I WILL SHO
      W YOU EACH INPUT DATUM."
60020 PRINT : PRINT " IF IT IS
      ";; INVERSE : PRINT "WRONG"
      ;: NORMAL : PRINT " YOU PRES
      S ";; FLASH : PRINT "W";: NORMAL
      : PRINT "."
60030 PRINT " OTHER WISE PRESS A
      NY KEYS."
60040 HOME
60050 FOR CN = 1 TO 2:VT = 12 *
      (CN - 1) + 1
60060 : VTAB VT: HTAB 13
60070 : PRINT " VIEW NUMBER ";; INVERSE : PRINT CN
60080 : HTAB 10: VTAB 3: PRINT "U
      ";; HTAB 30: PRINT "V"
60090 : NORMAL
60095 ::VT = VT + 4 / CN
60100 : FOR I = 0 TO 6
60110 : INVERSE
60120 : VTAB VT + I: PRINT I + 1;
      : NORMAL
60130 :: HTAB 4: PRINT TU(CN,I);:
      HTAB 24: PRINT TV(CN,I)
60140 NEXT I,CN
60200 : FOR CN = 1 TO 2
60210 ::VT = 12 * CN - 11 + 4 / C
      N
60220 :: FOR I = 0 TO 6
60230 :: VTAB VT + I: HTAB 4: GET
      CH$
60240 :: IF CH$ = "W" THEN GOSUB
      61240
60250 :: VTAB VT + I: HTAB 24: GET
      CH$
60260 :: IF CH$ = "W" THEN GOSUB
      61260
60270 :: NEXT I
60280 NEXT CN
60285 PRINT
60290 INVERSE : PRINT "ALL RIGH
      T?";: NORMAL : PRINT "YES O
      R NO";: GET CH$
60300 IF CH$ = "N" THEN 60040
60310 PRINT : RETURN
61240 INPUT " ";TU(CN,I)
61250 VTAB VT + I: HTAB 24: PRINT
      TV(CN,I): RETURN
61260 INPUT " ";TV(CN,I): RETURN

```

6.2.3 เปลี่ยนนิกิตทางเดินบนแต่ละภาพให้เป็นระบบนิกิตบนห้องฟอง มีขั้นตอนดังนี้

1. จองตัวแปรที่ใช้ในการเปลี่ยนระบบนิกิต (U, V) เป็น (X, Y)
2. เปลี่ยนนิกิต (U, V) เป็น (X, Y) ที่ละคู่จากแต่ละภาพ แล้วเก็บในตัวแปรที่

จองไว้จนครบทุกจุดทั้งสองภาพโดยใช้โปรแกรมย่อยช่วย

3. จองตัวแปรที่ใช้ในการหมุนนิกิตให้ขนานกับนิกิตบนห้องฟอง
4. เปลี่ยนนิกิต ตามข้อ 2 เป็นนิกิตที่มีแกนขนานกับแกนนิกิตของห้องฟองทุกจุด

จากแต่ละภาพ

5. จองตัวแปรที่ใช้ในการเลื่อนแกนนิกิตให้เป็นแกนเดียวกับแกนของห้องฟอง
6. เลื่อนแกนนิกิตของทุกจุดจากแต่ละภาพ


6.2.3.1 ความหมายของตัวแปร

ตัวแปร	$TX(CN, I), (CN, I)$	หมายถึงทางเดินอนุภาค (Track) โดยนิกิต X, Y
ตัวแปร	$RX(CN, I), RY(CN, I)$	หมายถึงนิกิต (X, Y) ที่หมุน (Rotate) ไปให้ขนานกับนิกิตของห้องฟอง
ตัวแปร	AL	หมายถึงมุมที่ใช้คำนวณการหมุนแกน
ตัวแปร	XX, YY	หมายถึงนิกิต X, Y ที่จะนำไปหมุน
ตัวแปร	XR, YR	หมายถึงนิกิต X, Y ที่ผ่านการหมุนแกนแล้ว

ศูนย์วิทยทรัพยากร
จุฬาลงกรณ์มหาวิทยาลัย

6.2.3.2 ผังงานและโปรแกรม


ผังงานย่อย



รูปที่ 51

ลำดับการทำงานย่อยเปลี่ยน (U,V) เป็น (X,Y) และลำดับการทำงานย่อยหมวกแทน

ผังงานหลัก



รูปที่ 52

ลำดับการทำงานหลักเปลี่ยนระบบแกนจาก (U,V) เป็น (x,y) แล้วหมุนแกนให้ขนานกับระบบ
ของจุดอ้างอิง และเลื่อนจุดกำเนิดให้เป็นระบบเดียวกับระบบจุดอ้างอิง

โปรแกรมย่อย

```

99  REM (U,V) TO (X,Y)
100 SO = (U + V + W) / 2
110 SU = SO - U
120 SV = SO - V
130 SW = SO - W
140 X = 2 * SO * SV / W - U
150 Y = - 2 * SQR (SO * SU * SV
      * SW) / W
160  RETURN
170  REM *****
190  REM ROTATE AXIS
200  XR = XX * COS (AL) - YY * SIN
      (AL)
210  YR = XX * SIN (AL) + YY * COS
      (AL)
220  RETURN

```

รูปที่ 53

โปรแกรมย่อยเปลี่ยนระบบ (U,V) เป็น (x,y) และโปรแกรมย่อยหมุนแกน

โปรแกรมหลัก

```

20610  REM BEFORE ANYTHING!
20611  REM CHANGE (U,V) TO (X,Y)
20620  DIM TX(2,6),TY(2,6)
20630  FOR CN = 1 TO 2
20640  : FOR I = 0 TO 6
20650  ::U = (TU(CN,I) - B1) / 40
20655  ::V = (B2 - TV(CN,I)) / 40
20660  : GOSUB 100
20670  ::TX(CN,I) = X
20675  ::TY(CN,I) = Y
20680  : NEXT I
20690  NEXT CN

```

รูปที่ 54

โปรแกรมหลักเปลี่ยนระบบพิกัด (U,V) เป็น (X,Y) แล้วโปรแกรมหมุนแกนให้แกนแกนของระบบของห้องฟอง และโปรแกรมเลื่อนแกนให้เป็นระบบเดียวกับพิกัดของจุดอ้างอิงในห้องฟอง

```
20691 REM *****
20692 PRINT "ROTATE TO PARALLEL
"
20693 REM *****
20700 DIM RX(2,6),RY(2,6)
20710 FOR CN = 1 TO 2
20720 :AL = RA(CN)
20730 : FOR I = 0 TO 6
20740 ::XX = TX(CN,I)
20745 ::YY = TY(CN,I)
20750 : GOSUB 200
20760 ::RX(CN,I) = XR
20765 ::RY(CN,I) = YR
20770 : NEXT I
20780 NEXT CN
20790 PRINT "SET THE SAME ORIGIN
"
20800 DIM MX(2,6),MY(2,6)
20810 FOR CN = 1 TO 2
20820 : FOR I = 0 TO 6
20830 ::MX(CN,I) = RX(CN,I) + X0(
CN)
20840 ::MY(CN,I) = RY(CN,I) + Y0(
CN)
20850 : NEXT I
20860 NEXT CN
20890 REM *****
```

รูปที่ 54 (ต่อ)

ศูนย์วิทยทรัพยากร
จุฬาลงกรณ์มหาวิทยาลัย

6.2.4 สร้างฟังก์ชันพหุนามของภาพถ่ายทางเดินจากกล้องที่ 1 มีขั้นตอนดังนี้


1. คำนวณมุมที่จะหมุนไปให้เหมาะสม โดยข้อมูลจากภาพที่ 1
2. จองตัวแปรเพื่อเปลี่ยนพิกัด ให้เหมาะแก่การทำสมการถดถอยพหุนาม
3. เปลี่ยนระบบพิกัดของตำแหน่งบนทางเดินอนุภาค โดยหมุนมุมตามข้อ 1
4. เปลี่ยนระบบพิกัดของตำแหน่งเลนส์บนภาพแต่ละภาพเช่นเดียวกับข้อ 3 ของทุกจุด และทั้ง 2 ภาพ (ใช้มุมหมุนแกนมุมเดียวกัน)
5. หาสัมประสิทธิ์ของการถดถอยพหุนาม โดยวิธีกำลังสองต่ำสุด


6.2.4.1 ความหมายของตัวแปร

ตัวแปร	BE	หมายถึงมุมหมุนที่คำนวณได้ (β)
ตัวแปร	XR(CN, I), YR(CN, I)	หมายถึงพิกัดของจุดบนทางเดินของอนุภาคที่หมุนให้เหมาะจะทำการถดถอย
ตัวแปร	LX(CN), LY(CN)	หมายถึงพิกัดของเลนส์ของกล้องที่ CN
ตัวแปร	A(I, J)	เป็นสัมประสิทธิ์ของระบบสมการเชิงเส้นที่ใช้ในการทำการถดถอยเชิงเส้น
ตัวแปร	B(I)	เป็นสัมประสิทธิ์ของฟังก์ชันพหุนามที่ได้

6.2.4.2 ฟังก์ชันและโปรแกรม

ฟังก์ชัน






5

รูปที่ 55

ลำดับขั้นการทำงานสร้างฟังก์ชันแทน

ผังงานย่อย



ลำดับการทำงานย่อยปรับไม่ให้ค่าในแนวเส้นทะแยงมุมของเมทริกซ์เป็น 0

โปรแกรม

```

20900 PRINT "ROTATE FOR REGRESSI
      ON"
20902 PRINT "SOLVE FOR ANGLE"
20910 DX = MX(1,6) - MX(1,0)
20920 DY = MY(1,6) - MY(1,0)
20930 PI = 3.141592653
20940 IF DX = 0 THEN BE = PI / 2
      : GOTO 20970
20950 BE = ATN (DY / DX): IF DX >
      0 THEN 20980
20960 BE = BE + PI: GOTO 20980
20970 IF DY < 0 THEN BE = - PI /
      2
20980 AL = - BE
20990 PRINT "ROTATING"
21000 DIM XR(2,6),YR(2,6)
21010 FOR CN = 1 TO 2
21020 : FOR I = 0 TO 6
21030 ::XX = MX(CN,I)
21040 ::YY = MY(CN,I)
21050 : GOSUB 200
21060 ::XR(CN,I) = XR
21070 ::YR(CN,I) = YR
21080 : NEXT I
21090 NEXT CN
21091 DIM LX(2),LY(2)
21092 FOR CN = 1 TO 2
21093 ::XX = XL(CN):YY = YL(CN)
21094 : GOSUB 200
21095 ::LX(CN) = XR:LY(CN) = YR
21096 NEXT CN
21100 REM *****
21101 FOR CN = 1 TO 2: FOR I = 0
      TO 6: PRINT XR(CN,I),YR(CN,
      I): NEXT I,CN
21110 REM POLYNOMIAL REGRESSION
21120 DIM A(5,5),B(5),X(10)
21121 : REM SET LINEAR EQUATION
21130 : FOR R = 0 TO 5
21140 :: FOR I = 0 TO 6
21150 :::B(R) = B(R) + (XR(1,I) /
      100) ^ R * (YR(1,I) / 100)
21160 :: NEXT I
21170 : NEXT R
21180 ::X(0) = 7
21190 : FOR I = 1 TO 10
21200 :: FOR J = 0 TO 6
21210 :::X(I) = X(I) + (XR(1,J) /
      100) ^ I

```

```

21220 :: NEXT J
21230 : NEXT I
21232 : REM SET MATRIX ELEMENT
21240 : FOR R = 0 TO 5
21250 :: FOR C = 0 TO 5
21260 :::A(R,C) = X(R + C)
21270 :: NEXT C
21280 : NEXT R
21300 REM SOLVE LINEAR EQUATION
21310 : FOR D = 0 TO 5
21320 :: IF A(D,D) = 0 THEN GOSUB
        62000
21330 :: FOR R = D TO 5
21340 ::: IF A(R,D) = 0 THEN 2144
        0
21360 ::::B(R) = B(R) / A(R,D)
21370 ::: IF R = D THEN 21390
21380 ::::B(R) = B(R) - B(D)
21390 :: FOR C = 5 TO D + 1 STEP
        - 1
21400 ::::A(R,C) = A(R,C) / A(R,D)
        )
21410 ::: IF R = D THEN 21430
21420 ::::A(R,C) = A(R,C) - A(D,C)
        )
21430 :: NEXT C
21440 :: NEXT R
21450 : NEXT D
21490 REM FINISH UPPER
21500 REM TRIANGULAR MATRIX
21510 : FOR R = 4 TO 0 STEP - 1
21520 :: FOR C = R + 1 TO 5
21530 ::::B(R) = B(R) - A(R,C) * B
        (C)
21540 :: NEXT C
21550 : NEXT R
21600 : FOR I = 0 TO 5
21610 ::B(I) = 100 * B(I) / 100 ^
        I
21620 : NEXT I
21630 REM *****
21640 REM FINISH REGRESSION

```

โปรแกรมย่อย

```

61999 REM SET A(C,C) <> 0
62000 ::JJ = D
62010 ::JJ = JJ + 1
62020 : IF JJ > 5 THEN PRINT "NO
        SOLUTION.": END
62030 : IF A(JJ,D) = 0 THEN 62010

62040 : FOR K = D TO 5
62050 ::A(D,K) = A(D,K) + A(JJ,K)

62055 ::B(D) = B(D) + B(JJ)
62060 : NEXT K
62070 RETURN : REM TO 21320

```

รูปที่ 58

โปรแกรมย่อยจัดค่าแนวเส้นทะแยงมุมของเมทริกซ์ไม่ให้เป็น 0

6.2.5 หาดำแหน่งบนทางเดินอนุภาคใน 3 มิติ

มีขั้นตอนต่าง ๆ ดังนี้

1. จอตัวแปรที่ใช้เก็บสัมประสิทธิ์ของฟังก์ชันผิวกรวย
2. สร้างสัมประสิทธิ์ของฟังก์ชันผิวกรวย
3. จอตัวแปรที่ใช้เก็บสัมประสิทธิ์ของฟังก์ชันจุดตัดของผิวกรวยกับเส้นตรงจาก

อีกภาพ

4. สร้างฟังก์ชัน, อนุพันธ์อันดับหนึ่ง และอนุพันธ์อันดับสองของฟังก์ชันจุดตัด
5. หาดำแหน่งใกล้สุดของเส้นที่ลากผ่านจุดเริ่มต้นทางเดินอนุภาค
6. การทำซ้ำโดยวิธีประยุกต์ของนิวตัน-รานส์ค้นหาคำตอบของฟังก์ชันในข้อ 4
7. หาดำแหน่งใน 3 มิติ
8. เปลี่ยนระบบพิกัดให้กลับเป็นระบบของห้องฟอง

6.2.5.1 ความหมายของตัวแปร


ตัวแปร	DL(1)	หมายถึง ผลต่างตำแหน่งเลนส์ในแกน X
ตัวแปร	DL(2)	หมายถึง ผลต่างตำแหน่งเลนส์ในแกน Y
ตัวแปร	DD = DL(2)/DL(1)	

ตัวแปร	RZ = Z(1)/Z(2)	
ตัวแปร	RL(1, I)	หมายถึงสัมประสิทธิ์ของฟังก์ชันผิวกรวยที่เกี่ยวกับแกน X
ตัวแปร	RL(2, I)	หมายถึงสัมประสิทธิ์ของฟังก์ชันผิวกรวยที่เกี่ยวกับแกน Y
ตัวแปร	IT(I, J)	หมายถึงสัมประสิทธิ์ของฟังก์ชันกำลังที่ I จุดตัดที่ J
ตัวแปร	FN FO(B)	หมายถึงฟังก์ชันจุดตัด
ตัวแปร	FN F1(B)	หมายถึงอนุพันธ์อันดับ 1 ของฟังก์ชันจุดตัด
ตัวแปร	FN F2(B)	หมายถึงอนุพันธ์อันดับ 2 ของฟังก์ชันจุดตัด
ตัวแปร	DI(I, J)	หมายถึงทิศทางของเส้นตรงที่ลากโดยกล้องที่ I แกนที่ J
ตัวแปร	CO(I, J)	หมายถึงจุดคงที่บนเส้นตรงที่ลากโดยกล้องที่ I แกนที่ J
ตัวแปร	T(I)	หมายถึง พารามิเตอร์ของสมการเส้นตรง ณ จุดตัด ของเส้นตรง ของภาพที่ I
ตัวแปร	L1(I)	หมายถึงตำแหน่งจุดตัดบนเส้นตรงของภาพที่ 1 แกนที่ I
ตัวแปร	L2(I)	หมายถึงตำแหน่งจุดตัดบนเส้นตรงของภาพที่ 2 แกนที่ I
ตัวแปร	LO(I)	หมายถึงตำแหน่งเฉลี่ยของจุดตัด แกนที่ I
ตัวแปร	ZZ(I)	หมายถึงพารามิเตอร์ที่ใช้หาจุดตัดใน 3 มิติ ซึ่งเป็นสัดส่วนกลับกับตำแหน่งในแกน Z
ตัวแปร	FC =	ค่าฟังก์ชัน FN FO(B)
ตัวแปร	F1 =	ค่าฟังก์ชัน FN F1(B)
ตัวแปร	F2 =	ค่าฟังก์ชัน FN F2(B)
ตัวแปร	IT =	เป็นตัวนับจำนวนครั้งของการทำซ้ำ
ตัวแปร	XT(I), YT(I), ZT(I)	หมายถึงพิกัดจุดตัดผิวกรวยตามแกน X, Y, Z



6.2.5.2 ผังงานและโปรแกรม

ผังงาน



รูปที่ 59

ลำดับการทำงานหาค่าแทนใน 3 มิติ

โปรแกรม

```

22030 DIM DL(2),RL(2,6)
22040 :DL(1) = (LX(2) - LX(1)) *
      Z(1)
22050 :DL(2) = (LY(2) - LY(1)) *
      Z(1)
22060 :DD = DL(2) / DL(1)
22070 :RZ = Z(1) / Z(2)
22080 FOR I = 0 TO 6
22090 ::RL(1,I) = (XR(2,I) - LX(2)
      ) * RZ + LX(1)
22100 ::RL(2,I) = (YR(2,I) - LY(2)
      ) * RZ + LY(1)
22110 NEXT I
22120 DIM IT(5,6)
22121 REM *****
22122 REM ITERATIVE FUNCTION
22123 REM *****
22130 FOR I = 0 TO 6
22140 : FOR J = 0 TO 5
22150 ::IT(0,I) = IT(0,I) + B(J)
      * RL(1,I) ^ J
22160 : NEXT J
22170 ::IT(0,I) = IT(0,I) - RL(2,
      I)
22180 : FOR J = 1 TO 5
22190 ::IT(1,I) = IT(1,I) + J *
      B(J) * RL(1,I) ^ (J - 1)
22200 : NEXT J
22210 ::IT(1,I) = IT(1,I) - DD
22220 : FOR J = 2 TO 5
22230 ::IT(2,I) = IT(2,I) + J *
      (J - 1) / 2 * B(J) * RL(1,I)
      ^ (J - 2)
22240 : NEXT J
22250 ::IT(3,I) = B(3) + 4 * B(4)
      * RL(1,I) + 10 * B(5) * RL(
      1,I) ^ 2
22260 ::IT(4,I) = B(4) + 5 * B(5)
      * RL(1,I)
22270 ::IT(5,I) = B(5)
22275 NEXT I
22400 DEF FN F0(B) = IT(0,P) +
      IT(1,P) * B + IT(2,P) * B ^
      2 + IT(3,P) * B ^ 3 + IT(4,P)
      * B ^ 4 + IT(5,P) * B ^ 5
22410 DEF FN F1(B) = IT(1,P) +
      2 * IT(2,P) * B + 3 * IT(3,P)
      * B ^ 2 + 4 * IT(4,P) * B ^
      3 + 5 * IT(5,P) * B ^ 4

```

```

22420 DEF FN F2(B) = 2 * IT(2,P
) + 6 * IT(3;P) * B + 12 * I
T(4,P) * B ^ 2 + 20 * IT(5,P
) * B ^ 3
22490 REM *****
22491 REM # INTERSECTION OR #
22492 REM # THE CLOESEST POINT#
22493 REM *****
22500 DIM DI(2,3),CO(2,3)
22510 FOR I = 1 TO 2
22520 ::DI(I,1) = XR(I,0) - LX(I)

22530 ::DI(I,2) = YR(I,0) - LY(I)

22540 ::DI(I,3) = Z(I)
22550 ::CO(I,1) = LX(I)
22560 ::CO(I,2) = LY(I)
22570 ::CO(I,3) = 0
22580 NEXT I
22590 FOR I = 1 TO 3
22600 ::CO(0,I) = CO(1,I) - CO(2,
I)
22610 NEXT I
22620 ::DD = 0:D1 = 0:D2 = 0:DC =
0:CD = 0
22630 FOR I = 1 TO 3
22640 ::DD = DD + DI(1,I) * DI(2,
I)
22650 ::D1 = D1 + DI(1,I) * DI(1,
I)
22660 ::D2 = D2 + DI(2,I) * DI(2,
I)
22670 ::DC = DC + DI(1,I) * CO(0,
I)
22680 ::CD = CD + DI(2,I) * CO(0,
I)
22690 NEXT I
22700 DIM T(2)
22710 ::T(0) = D1 * D2 - DD * DD
22720 ::T(1) = CD * DD - DC * D2
22730 ::T(2) = CD * D1 - DC * DD
22740 ::T(1) = T(1) / T(0)
22750 ::T(2) = T(2) / T(0)
22760 DIM L0(3),L1(3),L2(3)
22770 FOR I = 1 TO 3
22780 ::L1(I) = DI(1,I) * T(1) +
CO(1,I)
22790 ::L2(I) = DI(2,I) * T(2) +
CO(2,I)
22800 ::L0(I) = L1(I) / 2 + L2(I)
/ 2
22810 NEXT I

```

```

22820 DIM ZZ(6)
22830 REM *****
22840 REM * START ITERATION *
22850 REM *****
22890 P0 = DL(1) / L0(3)
22900 FOR P = 0 TO 6
22910 ::IT = 0
22920 ::F0 = FN F0(P0)
22930 :: IF F0 = 0 THEN 23020
22940 ::F1 = FN F1(P0)
22950 ::F2 = FN F2(P0)
22960 ::P1 = P0 - F0 * F1 / (F1 *
      F1 - F0 * F2)
22970 :: IF ABS ((P1 - P0) / P1)
      < 1E - 9 THEN 23020
22980 ::P0 = P1
22990 ::IT = IT + 1
23000 :: IF IT > 50 THEN PRINT "
      NO SOLUTION": END
23010 :: GOTO 22920
23020 :ZZ(P) = P1
23030 P0 = P1
23040 NEXT P
23050 REM *****
23060 REM * END OF ITERATION *
23070 REM *****
23100 REM
23200 REM *****
23210 REM * POSITION ON TRACK *
23220 REM * IN THREE DIMENSION*
23230 REM *****
23240 DIM ZT(6),YT(6),XT(6)
23250 FOR I = 0 TO 6
23260 ::ZT(I) = DL(1) / ZZ(I)
23270 ::XT(I) = LX(2) + ZT(I) / Z
      (2) * (XR(2,I) - LX(2))
23280 ::YT(I) = LY(2) + ZT(I) / Z
      (2) * (YR(2,I) - LY(2))
23290 NEXT I
23299 REM *****
23300 REM TURN TO FID. SYS.
23301 ::AL = - AL
23302 FOR I = 0 TO 6
23303 ::XX = XT(I):YY = YT(I)
23304 : GOSUB 200
23305 ::XT(I) = XR:YT(I) = YR
23306 ::ZT(I) = FZ(1) - ZT(I)
23307 NEXT I

```


6.2.6 หาค่าสำคัญของทางเดินอนุภาค
มีขั้นตอนตามนี้


1. หาจุดศูนย์กลางของวงกลม และรัศมี
2. หามุม ชีนิท (Zenith)
3. หามุม อาซิมุท (Azimuth)
4. แสดงค่าที่หาได้

6.2.6.1 ความหมายของตัวแปร

ตัวแปร	$RO(I)$	=	รัศมีโดยการคำนวณครั้งที่ I
ตัวแปร	$XO(I), YO(I)$	=	พิกัดของจุดศูนย์กลางวงกลมคำนวณครั้งที่ I
ตัวแปร	ZE	=	มุมชีนิท
ตัวแปร	AZ	=	มุมอาซิมุท

6.2.6.2 ผังงานและโปรแกรม

ผังงาน



รูปที่ 61

ลำดับการทำงานหาค่าตอบต่างจากทางเดินใน 3 มิติ

โปรแกรม

```

23309 DIM RO(5),XO(5),YO(5)
23310 REM *****
23320 REM * SERCHING CENTER *
23330 REM *****
23340 FOR J = 1 TO 5
23350 I = J - 1
23360 K = J + 1
23370 A = XT(J) - XT(I)
23380 B = YT(J) - YT(I)
23390 C = XT(K) - XT(J)
23400 D = YT(K) - YT(J)
23410 E = (XT(J) ^ 2 + YT(J) ^ 2 -
XT(I) ^ 2 - YT(I) ^ 2) / 2
23420 F = (XT(K) ^ 2 + YT(K) ^ 2 -
XT(J) ^ 2 - YT(J) ^ 2) / 2
23430 DE = (A * D - C * B)
23440 XO = (E * D - F * B) / DE
23450 YO = (A * F - C * E) / DE
23460 RO = SQR ((XT(J) - XO) ^ 2
+ (YT(J) - YO) ^ 2)
23470 RO(J) = RO
23480 XO(J) = XO
23490 YO(J) = YO
23500 NEXT J
23530 RO = RO(1)
23540 XO = XO(1)
23550 YO = YO(1)
23600 REM *****
23610 REM * INCLINATION ANGLE *
23620 REM * OR ZENITH ANGLE *
23630 REM *****
23640 :DX = XT(1) - XT(0)
23650 :DY = YT(1) - YT(0)
23660 :SS = DX ^ 2 + DY ^ 2
23670 ::S = SQR (SS)
23680 :TH = ATN (S / SQR (4 * R
0 ^ 2 - SS))
23690 :KK = TH / (ZT(1) - ZT(0))
23700 :TZ = 1 / KK / RO
23710 :ZE = PI / 2 - ATN (TZ)

```

รูปที่ 62

โปรแกรมการหาค่าตอบต่าง ๆ จากทางเดินใน 3 มิติ

```

24300 REM *****
24310 PRINT "START TRACK AT (X,Y
) = ("XT(O)","YT(O))"
24320 PRINT "CENTER (X,Y) = (";X
0;","YO")"
24330 PRINT "ZENITH ANGLE OF TAN
GENT AT STARTING POINT I
S "ZE" RADIAN."
24340 :AX = XT(O) - X0:AY = YT(O)
- Y0
24350 IF AX = 0 THEN AZ = PI / 2
: GOTO 24400
24360 :AZ = ATN (AY / AX): IF AX
> 0 THEN 24410
24370 :AZ = AZ + PI: GOTO 24410
24400 IF AY < 0 THEN AZ = - PI /
2
24410 AZ = AZ + PI / 2
24420 PRINT "AZIMUTH ANGLE OF TA
NGENT AT STARTING POINT I
S "AZ" RADIAN."
59990 END
59999 REM *****

```

รูปที่ 62 (ต่อจบ)

ศูนย์วิทยทรัพยากร
จุฬาลงกรณ์มหาวิทยาลัย